```
$regfile = "attiny2313.dat"
$crystal = 500000 '4 MHz Clock & CLKDIV/8
$hwstack = 32
$swstack = 10
$framesize = 40

CONFIG SUBMODE = NEW

CONST true = 1
CONST false = 0
CONST pressed = 0
CONST LEDoff = 1
CONST LEDon = 0

CONST pwmUPfast = 2
CONST pwmUPslow = 3
CONST pwmDNslow = 4
CONST pwmDNfast = 5
CONST pwmRTcw = 6
CONST pwmRTccw = 7
CONST Timer0Preset = 176

DIM Event10ms AS BYTE
DIM pwm AS BYTE
DIM pwmcntr AS BYTE
DIM x AS BYTE
DIM k as BYTE

'************************************************
btnUPfast ALIAS PINB.7
btnUPslow ALIAS PINB.6
btnDNslow ALIAS PINB.5
btnDNfast ALIAS PINB.4
pwmOut ALIAS PORTB.3
btnRTcw ALIAS PINB.1
btnRTccw ALIAS PINB.0
LEDgreen ALIAS PORTD.2
LEDred ALIAS PORTD.3

'Frequency
Config btnUPfast = INPUT : btnUPfast = 1
Config btnUPslow = INPUT : btnUPslow = 1
Config btnDNslow = INPUT : btnDNslow = 1
Config btnDNfast = INPUT : btnDNfast = 1

'Rotation
Config btnRTcw = INPUT : btnRTcw = 1
Config btnRTccw = INPUT : btnRTccw = 1

'PWM Command
Config pwmOut = OUTPUT

'LED
```

```
Config LEDgreen = OUTPUT
Config LEDred = OUTPUT

'#########################################
' FUNCTIONS
'#########################################
LEDgreen = LEDon
LEDred = LEDoff
pwmOut = false

' Blink LED for Startup Welcome Message
x=10
While x>0
  Toggle LEDgreen
  waitms 100
  DECR x
Wend
waitms 250
x=9
While x>0
  Toggle LEDred
  waitms 100
  DECR x
Wend

'WatchDog
'Set Timeout period   => 500ms
WDTCR.WDP0 = 1
WDTCR.WDP1 = 0
WDTCR.WDP2 = 1
WDTCR.WDP3 = 0
'Enable Watchdog
WDTCR.WDIE = 0
WDTCR.WDCE = 1
WDTCR.WDE = 1

'Init Timer0 (10ms cycle)
CONFIG TIMER0 = TIMER , PRESCALE = 64
START TIMER0
ON OVF0 isrTimer0_Overflow
Enable OVF0
Enable Interrupts

Do
  If Event10ms = true Then
    Event10ms = false
    RESET Watchdog

    'Check Buttons
    k = PINB AND 0b11110011
    Select Case k
      Case 0b11110011 'no button pressed
        pwm = 0
```

```
      Case 0b11110010 'btnRTccw
        pwm = pwmRTccw
      Case 0b11110001 'btnRTcw
        pwm = pwmRTcw
      Case 0b11100011 'btnDNfast
        pwm = pwmDNfast
      Case 0b11010011 'btnDNslow
        pwm = pwmDNslow
      Case 0b10110011 'btnUPslow
        pwm = pwmUPslow
      Case 0b01110011 'btnUPfast
        pwm = pwmUPfast
      Case Else
        pwm = 0
    End Select
  End If
Loop

'#############################################################################

'Timer0
isrTimer0_Overflow:
  TIMER0 = Timer0Preset
  Select Case pwmcntr
    Case 0
      If pwm > 0 Then
        pwmOut = true
        Incr pwmcntr
      End If
    Case Is >8
      pwmOut = false
      pwmcntr = 0
    Case Else
      If pwmcntr = pwm Then
        pwmOut = false
      End if
      Incr pwmcntr
  End Select
  LEDred = NOT pwmOut
  Event10ms = true
Return
```